NCL

Part VII

Caption and Annotation

Exercises and Tasks

NCL

Caption and Annotations

DKRZ NCL workshop: Text settings

Umlaute: Ä ä Ö ö Ü ü

Superscript: m^3 [m s^{-1}] [kg m^{-2}] 30°C

Subscript: Schwefelsäure: H_2SO_4

Format: 1. 16.25 0.0
2. -0.93 0.0
3. 0.43 0.0
4. 3.50 0.0

Greek font: alpha = α beta
sigma = σ

Font size 100% Font size 70%

Helvetica-bold font size 0.03



NCL

Map annotations



Map annotations (1/4)

```
begin
  f    =  addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
  var =  f->tsurf(0,:,:)

  wks = gsn_open_wks("png","plot_part_VII_map_annotation")

  mres1                  =  True
  mres1@gsnDraw          =  False                ; don't draw the plot yet
  mres1@gsnFrame         =  False                ; don't advance the frame
  mres1@gsnMaximize      =  True                 ; maximize the plot output

  mres1@cnFillOn         =  True                 ; filled contours
  mres1@cnLinesOn        =  False                ; don't draw contour lines

  mres1@mpDataBaseVersion = "MediumRes"          ; set map data base
  mres1@mpOutlineOn      =  True                 ; turn map outline on
  mres1@mpGeophysicalLineThicknessF = 2          ; increase coastline thickness
  mres1@mpMinLonF        = -30.                  ; min lon
  mres1@mpMaxLonF        =  30.                  ; max lon
  mres1@mpMinLatF        =  30.                  ; min lat
  mres1@mpMaxLatF        =  80.                  ; max lat
```


Map annotations (2/4)

```
  mres1@lbBoxMinorExtentF     =  0.15            ; decrease height of labelbar

  mres1@tiMainString       = "DKRZ NCL workshop:  map annotation" ; title string
  mres1@tiMainFontHeightF     =  0.02            ; main title font size

  map1 = gsn_csm_contour_map(wks, var, mres1)    ; draw the contour map

  mres2                       =  True            ; resources smaller world map
  mres2@gsnDraw               =  False           ; don't draw the plot yet
  mres2@gsnFrame              =  False           ; don't advance the frame
  mres2@gsnLeftString         =  ""              ; don't draw left string
  mres2@gsnRightString        =  ""              ; don't draw right string

  mres2@vpWidthF              =  0.3             ; set viewport width
  mres2@vpHeightF             =  0.3             ; set viewport height

  mres2@mpLandFillColor       = "grey75"         ; land fill color
  mres2@mpOceanFillColor      = "white"          ; ocean fill color
  mres2@mpInlandWaterFillColor = "white"         ; inland water color
```


Map annotations (3/4)

```
  mres2@tmXBOn                =  False           ; turn off tickmarks at bottom
  mres2@tmXTOn                =  False           ; turn off tickmarks at top
  mres2@tmYLOn                =  False           ; turn off tickmarks at left side
  mres2@tmYROn                =  False           ; turn off tickmarks at right side
  map2 = gsn_csm_map(wks, mres2)                 ; create the smaller world map,
                                                 ;        but don't draw it yet
  pres                        =  True            ; set polyline resources
  pres@gsLineThicknessF       =  2.0             ; polyline thickness
  pres@gsLineColor            = "blue"           ; polyline color

;-- attach polyline to map2
  pbox = gsn_add_polyline(wks, map2,(/-30,30,30,-30,-30/),(/30,30,80,80,30/),pres)

  amres                       =  True            ; set annotation resources
  amres@amJust                = "TopLeft"        ; labelbar justification
  amres@amParallelPosF        = -0.5            ; move labelbar to the right of plot
  amres@amOrthogonalPosF      = -0.5            ; move labelbar to the top of plot

  annoid = gsn_add_annotation(map1,map2,amres)   ; atach the labelbar to the plot
```

NCL

Map annotations (4/4)

```
  txres1                  =  True             ; set text resources
  txres1@txFontHeightF    =  0.03             ; smaller text font size
  txres1@txAngleF         =  40.              ; rotate text string 40 degrees

  text1 = gsn_add_text(wks,map1,"Atlantic",-20,45,txres1) ; attach text to map1

  txres2                  =  True             ; set text resources
  txres2@txFontHeightF    =  0.03             ; smaller text font size
  txres2@txAngleF         =  -50.             ; rotate text string 40 degrees

  text2 = gsn_add_text(wks,map1,"North Sea",1.0,61,txres2) ; attach text to map1

;-- draw the map and advance the frame
  draw(map1)
  frame(wks)

end
```


NCL

Map annotations



NCL

Labelbar/legend annotation



NCL

Labelbar/legend annotation (1/4)

```
begin

  llab = (/"a","b","c","d","e","f","g","h","i","j"/)      ; define labels
  cols = (/"firebrick4","red","orange","yellow","green","darkgreen", \
           "cyan","DeepSkyBlue","blue","midnightblue"/)  ; define colors

  nlb  = dimsizes(cols)                                  ; number of labelbar boxes

  x    = (/1,2,3,4,5,6,7,8,9,10/)                        ; define x-values
  y    = (/32,25,35,21,28,31,11,19,16,14/)               ; define y-values

  wks = gsn_open_wks("png","plot_part_VII_legend_annotation")

;-- create XY bar chart plot
  res                       =  True
  res@gsnDraw               =  False          ; don't draw plot yet
  res@gsnFrame              =  False          ; don't advance frame yet
  res@gsnMaximize           =  True           ; maximize plot output
  res@gsnXYBarChart         =  True           ; turn on bar chart
  res@gsnXYBarChartBarWidth =  0.4            ; set width of bins
  res@gsnXYBarChartColors   =  cols           ; set the colors for the bins
```


NCL

Labelbar/legend annotation (2/4)

```
  res@trYMinF               =  0              ; start bins from bottom
  res@trYMaxF               =  max(y)+5       ; start bins from bottom
  res@trXMinF               =  0              ; start at x-value 0
  res@trXMaxF               =  11             ; end at x-value 11

  res@tiMainString = "DKRZ NCL workshop:  Labelbar Annotation"   ; title string

  plot = gsn_csm_xy (wks,x,y,res)          ; create the plot, but don't draw it yet

  getvalues plot
       "vpWidthF"  :  vpw                  ; get viewport width of plot
       "vpHeightF" :  vph                  ; get viewport height of plot
  end getvalues

;-- create labelbar
  lres                      =  True           ; set labelbar resources
  lres@lbAutoManage         =  True           ; necessary to control sizes
  lres@lbFillColors         =  cols           ; labelbar colors
  lres@vpWidthF             =  0.2 * vpw      ; labelbar width
  lres@vpHeightF            =  0.5 * vph      ; labelbar height
  lres@lbBoxMajorExtentF    =  0.7            ; insert white space between boxes
  lres@lbMonoFillPattern    =  True           ; solid fill pattern
```


NCL

Labelbar/legend annotation (3/4)

```
  lres@lbLabelFontHeightF   =  0.08           ; font height
  lres@lbLabelJust          = "CenterLeft"    ; label justification

  lbid = gsn_create_labelbar(wks,nlb,llab,lres)  ; create the labelbar,
                                                 ; but don't draw it yet
;-- add labelbar to plot
  amres                     =  True           ; set annotation resources
  amres@amJust              = "TopRight"      ; labelbar justification
  amres@amParallelPosF      =  0.5            ; move labelbar to the right of plot
  amres@amOrthogonalPosF    = -0.5            ; move labelbar to the top of plot

  annoid = gsn_add_annotation(plot,lbid,amres)   ; attach the labelbar to the plot

;-- create text
  txres                     =  True           ; set text resources
  txres@txPerimOn           =  True           ; draw a box around the text
  txres@txBackgroundFillColor = "cornsilk"    ; box fill color
  txres@txFontHeightF       =  0.03           ; smaller text font size

  txid = gsn_create_text(wks, "Text string, e.g. data description", txres)
                                                 ; create text, but don't draw it yet
```

NCL

Labelbar/legend annotation (4/4)

```
;-- add text annotation to plot
  amres@amJust           = "TopLeft"       ; labelbar justification
  amres@amParallelPosF   = -0.5            ; move labelbar to the right of plot
  amres@amOrthogonalPosF = -0.5            ; move labelbar to the top of plot

  annoid = gsn_add_annotation(plot, txid, amres)   ; attach text to the plot

;-- draw complete plot and advance the frame
  draw(plot)
  frame(wks)

end
```


NCL

Labelbar, legend annotation



NCL

Text settings

DKRZ NCL workshop: Text settings

Umlaute: Ä ä Ö ö Ü ü

Superscript: m^3 [m s^{-1}] [kg m^{-2}] 30°C

Subscript: Schwefelsäure: H_2SO_4

Format:

1.	16.25	0.06	2.0
2.	-0.93	0.02	2.4
3.	0.43	0.04	1.1
4.	3.50	0.05	0.9

Greek font: alpha = α beta = β
sigma = σ

Font size 100% Font size 70% Font size 40%

Helvetica-bold font size 0.03

Text settings (1/6)

```
begin

;-- define German "Umlaute"
  Auml = "A~H-15V6F35~H~FV-6H3~"              ; Ä
  auml = "a~H-13V2F35~H~FV-2H3~"              ; ä
  Ouml = "O~H-16V6F35~H~FV-6H3~"              ; Ö
  ouml = "o~H-14V2F35~H~FV-2H3~"              ; ö
  Uuml = "U~H-15V6F35~H~FV-6H3~"              ; Ü
  uuml = "u~H-13V2F35~H~FV-2H3~"              ; ü

;-- define super- and subscript character
  super = "m~S~3~N~  [m s~S~-1~N~]  [kg m~S~-2~N~]  30~S~o~N~C"
  sub   = "Schwefels" + auml + "ure:   H~B~2~N~SO~B~4~N~"

  data  = (/ "1.",      "2.",     "3.",     "4."/)
  diff  = (/ 16.25,    -0.93,     0.43,     3.5/)
  var   = (/  0.06,     0.02,     0.04,     0.05/)
  ratio = (/     2,      2.4,      1.1,      0.9/)

  ntext = dimsizes(data)
```


Text settings (2/6)

```
  wks = gsn_open_wks("png","plot_part_VII_text_settings")

;-- x, y start point for writing
  x   = 0.1
  y   = 0.95
  inc = 0.06

;-- text resources
  txres                = True
  txres@txFont         = "helvetica-bold"     ; change font
  txres@txFontHeightF  =  0.03                ; set font size
  txres@txJust         = "CenterCenter"       ; set text justification
  str = "DKRZ NCL workshop: Text settings"    ; title string
  gsn_text_ndc(wks,str,0.5,y,txres)           ; draw title string

  txres@txFont         = "helvetica"          ; change font
  txres@txJust         = "CenterLeft"         ; change text justification
  str1 = "Umlaute:"
  gsn_text_ndc(wks,str1,x,y-2*inc,txres)      ; draw string

  str2 = Auml+"  "+auml+"   "+Ouml+"  "+ouml+"   "+Uuml+" "+uuml
  gsn_text_ndc(wks,str2,x+0.3,y-2*inc,txres)  ; draw string
```


Text settings (3/6)

```
  str1 = "Superscript:"
  gsn_text_ndc(wks,str1,x,y-3*inc,txres)
  str2 = super                                ; definition of super see above
  gsn_text_ndc(wks,str2,x+0.3,y-3*inc,txres)

  str1 = "Subscript:"
  gsn_text_ndc(wks,str1,x,y-4*inc,txres)
  str2 = sub                                  ; definition of sub see above
  gsn_text_ndc(wks,str2,x+0.3,y-4*inc,txres)

;-- nice formated text output using sprintf
  str = "Format:"
  gsn_text_ndc(wks,str,x,y-5*inc,txres)

;-- draw data content
  xpos = 0.4
  do i=0,ntext-1
    ypos = y-5*inc-i*0.05
    gsn_text_ndc(wks,data(i),xpos,ypos,txres)
  end do
```

NCL

Text settings (4/6)

```
txres@txJust = "CenterRight"

do i=0,ntext-1
  xpos = 0.65                          ; column 1
  ypos = y-5*inc-i*0.05                ; position of rows
  str = sprintf("%6.2f",diff(i))       ; formated output of array diff
  gsn_text_ndc(wks,str,xpos,ypos,txres)

  xpos = xpos + 0.12                   ; column 2
  str = sprintf("%5.2f",var(i))        ; formated output of array var
  gsn_text_ndc(wks,str,xpos,ypos,txres)

  xpos = xpos + 0.12                   ; column 3
  str = sprintf("%3.1f",ratio(i))      ; formated output of array ratio
  gsn_text_ndc(wks,str,xpos,ypos,txres)
end do
```


NCL

Text settings (5/6)

```
;-- greek font
 xpos = 0.3
 ypos = 0.35

 str1 = "Greek font:"
 gsn_text_ndc(wks,str1,xpos,ypos,txres)

 str2 = "alpha = ~F33~a~N~"
 gsn_text_ndc(wks,str2,xpos+0.27,ypos,txres)

 str2 = "beta   = ~F33~b~N~"
 gsn_text_ndc(wks,str2,xpos+0.55,ypos,txres)

 str2 = "sigma = ~F33~s~N~"
 gsn_text_ndc(wks,str2,xpos+0.27,ypos-0.05,txres)
```


NCL

Text settings (6/6)

```
;-- change font size
 str1 = "Font size 100%"
 gsn_text_ndc(wks,str1,xpos+0.08,ypos-3*inc,txres)

 str2 = "~Z70~Font size 70%~N~"
 gsn_text_ndc(wks,str2,xpos+0.3,ypos-3*inc,txres)

 str3 = "~Z40~Font size 40%~N~"
 gsn_text_ndc(wks,str3,xpos+0.45,ypos-3*inc,txres)

;-- change Helvetica font to Helvetica-bold
 txres1                 =  True
 txres1@txFont          = "helvetica-bold"
 txres1@txFontHeightF   =  0.03
 txres1@txJust          = "CenterLeft"

 bold = "Helvetica-bold font size 0.03"
 gsn_text_ndc(wks,bold,x,0.08,txres1)

 frame(wks)                                ; advance the frame

end
```

NCL

Text settings

DKRZ NCL workshop: Text settings

Umlaute: Ä ä Ö ö Ü ü

Superscript: m^3 [m s^{-1}] [kg m^{-2}] 30°C

Subscript: Schwefelsäure: H_2SO_4

Format:

1.	16.25	0.06	2.0
2.	-0.93	0.02	2.4
3.	0.43	0.04	1.1
4.	3.50	0.05	0.9

Greek font: alpha = α beta = β
sigma = σ

Font size 100% Font size 70% Font size 40%

Helvetica-bold font size 0.03

Karin Meier-Fleischer, DKRZ 22

NCL

Task: Annotate map with text and add a legend

- Use the script **part_I_XY_multiple_curves.ncl**
- Modify:
  - Change the title string to "**Zonal winds**"
  - Set the **line thicknesses** to **3.0**
  - **Add a legend** for the two line graphs in the **lower right corner** with the labels "**-105° lon**" for the **red line** and "**10° lon**" for the **blue line**
  - **Add a text box** at the **lower left corner** of the plot containing the **units** of the data variable. The text box **background color** is **cornsilk**.

Hints:
- xyLineThicknesses/lgLineThicknessF
- gsn_create_legend
- u@units
- txBackgroundFillColor
- gsn_create_text
- gsn_add_annotation



Karin Meier-Fleischer, DKRZ 23

NCL

Task: Annotate map with text and legend



Karin Meier-Fleischer, DKRZ 24

Task: Annotate map with text and legend (1/3)

```
begin
   f = addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
   u = f->u10                                  ; set variable u

   data      = new((/2,dimsizes(u&lat)/), float)  ; assign multidimensional array
   data(0,:) = u(0,:,{10})                        ; values at longitude 10 deg.
   data(1,:) = u(0,:,{-105})                      ; values at longitude -105 deg.

   wks = gsn_open_wks("png", "task_VII_xy_plot_annotations")

   res                   =  True                ; create plot resource object
   res@gsnDraw           =  False               ; don't draw the plot yet
   res@gsnFrame          =  False               ; don't advance the frame

   res@xyLineColors      = (/"blue", "red"/)    ; line colors
   res@xyLineThicknesses = (/3.0,3.0/)          ; set line thicknesses

   res@tiMainString      = "Zonal winds"        ; draw title

   res@trYMinF           = -30.                 ; y-axis minimum
   res@trYMaxF           =  20.                 ; y-axis maximum

   plot = gsn_csm_xy(wks, u&lat, data, res)     ; create the plot, but don't draw it
```


Task: Annotate map with text and legend (2/3)

```
;-- set legend resources
  lres                    =  True
  lres@lgLineColors       =  (/"blue","red"/)
  lres@lgLineThicknessF   =  3.0
  lres@lgLabelFontHeightF =  .08              ; set the legend label font thickness
  lres@lgLabelJust        = "CenterRight"     ; set text justification

  lres@vpWidthF           =  0.2              ; width of legend in NDC coordinates
  lres@vpHeightF          =  0.075            ; height of legend in NDC coordinates

  lbid = gsn_create_legend(wks,2,(/"  10~S~o~N~ lon"," -105~S~o~N~ lon"/),lres)
                                              ; create the legend
;-- set annotation resources
  amres                  =  True
  amres@amJust           = "BottomRight"      ; legend justification
  amres@amParallelPosF   =  0.5               ; move legend to the right
  amres@amOrthogonalPosF =  0.5               ; move the legend down

  annoid1 = gsn_add_annotation(plot,lbid,amres)   ; attach legend to plot
```


Task: Annotate map with text and legend (3/3)

```
;-- set text resources
  txres                    =  True
  txres@txPerimOn          =  True            ; draw a box around the text
  txres@txBackgroundFillColor = "cornsilk"    ; box fill color
  txres@txFontHeightF      =  0.02            ; smaller text font size

  txid = gsn_create_text(wks, "units: "+u@units, txres)   ; create text, but
                                                          ; don't draw it yet

  amres@amJust             = "BottomLeft"     ; legend justification
  amres@amParallelPosF     = -0.5             ; move legend to the left of plot
  amres@amOrthogonalPosF   =  0.5             ; move legend downward

  annoid = gsn_add_annotation(plot, txid, amres) ; attach text to the plot

;-- draw the plot and advance the frame
  draw(plot)
  frame(wks)

end
```