NCL



Part III

Vectors

Exercises and Tasks

NCL

Vectors



NCL

Simple vector plot

```
begin

   f = addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
   u = f->u10(0,:,:)                        ; u-velocity, 1st time step
   v = f->v10(0,:,:)                        ; v-velocity, 1st time step

   wks =  gsn_open_wks("png","plot_part_III_simple_vectors")

   res               =  True                ; create resource object
   res@tiMainString  = "Simple vector plot" ; title string

   plot = gsn_csm_vector_map(wks,u,v,res)   ; create vector plot

end
```



Simple vector plot



Less vectors

```
begin

   f = addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
   u = f->u10(0,:,:)                        ; u-velocity, 1st time step
   v = f->v10(0,:,:)                        ; v-velocity, 1st time step

   wks =  gsn_open_wks("png","plot_part_III_less_vectors")

   res                  =  True             ; create resource object
   res@tiMainString     = "Less vectors"    ; title string

   res@vcRefMagnitudeF  =  5.               ; make vectors larger
   res@vcRefLengthF     =  0.025            ; ref vector length

   plot = gsn_csm_vector_map(wks,u(::4,::4),v(::4,::4),res)
                                            ; create vector plot

end
```


Less vectors



Vector plot settings (1/3)

```
begin
   f = addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
   u = f->u10(0,:,:)                       ; u-velocity, 1st time step
   v = f->v10(0,:,:)                       ; v-velocity, 1st time step

   wks =  gsn_open_wks("png","plot_part_III_vector_settings")

   res                  =  True            ; create resource object
   res@tiMainString     = "Vector plot settings"    ; draw a title

   res@vcRefMagnitudeF  =  5.              ; make vectors larger
   res@vcRefLengthF     =  0.02            ; reference vector length
   res@vcMinDistanceF   =  0.01            ; thin out vectors

   plot = gsn_csm_vector_map(wks,u,v,res)
```


Vector plot settings (2/3)



Vector plot settings (3/3)

```
   res@vcGlyphStyle  = "CurlyVector"        ; turn on curly vectors

   plot = gsn_csm_vector_map(wks,u,v,res)

end
```



NCL

Vector plot colorize (1/2)

```
begin
   f  =  addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
   u  =  f->u10(0,:,:)                  ; u-velocity, first time step
   v  =  f->v10(0,:,:)                  ; v-velocity, first time step

   wks =  gsn_open_wks("png","plot_part_III_vector_colorize")

   res                     =  True      ; create plot resource object
   res@tiMainString        = "Vector plot colorize" ;-- draw a title

   res@vcMinDistanceF      =  0.01      ; thin out vectors
   res@vcGlyphStyle        = "CurlyVector" ; turn on curly vectors
   res@vcLineArrowColor    = "white"    ; change vector color to white
   res@vcLineArrowThicknessF =  2.0     ; change vector thickness
   res@vcRefMagnitudeF     =  5.        ; reference vector magnitude
   res@vcRefLengthF        =  0.025     ; reference vector length
   res@vcRefAnnoArrowUseVecColor =  False  ; don't use vcLineArrowColor
                                           ; for reference vector
```


NCL

Vector plot colorize (2/2)

```
   res@mpOutlineOn       =  True              ; draw coastlines
   res@mpLandFillColor   = "AntiqueWhite2"    ; fill land areas color
   res@mpOceanFillColor  = "lightblue"        ; fill ocean areas color
   res@mpInlandWaterFillColor = "lightblue"   ; fill inland water color

   plot = gsn_csm_vector_map(wks,u,v,res)
end
```



NCL

Vector plot colorized by magnitude (1/2)

```
begin
   f =  addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
   u =  f->u10(0,:,:)                       ; u-velocity, 1st time step
   v =  f->v10(0,:,:)                       ; v-velocity, 1st time step
   mag       =  sqrt(u^2+v^2)               ; magnitude
   mag@units = "[m/s]"                      ; magnitude units

   wks =  gsn_open_wks("png","plot_part_III_vector_colorize_mag")

   res                     =  True          ; create resource object
   res@tiMainString        = "Vector plot colorized by magnitude"

   res@vcGlyphStyle        = "CurlyVector" ; turn on curly vectors
   res@vcMinDistanceF      =  0.01          ; thin out vectors
   res@vcLineArrowThicknessF =  2.0         ; change vector thickness
   res@vcRefMagnitudeF     =  5.            ; make vectors larger
   res@vcRefLengthF        =  0.025         ; ref vector length
   res@vcRefAnnoArrowUseVecColor =  False   ; ref vector in black
```

NCL

Vector plot colorized by magnitude (2/2)

```
res@mpOutlineOn             =  True          ; draw coastlines
res@mpLandFillColor         = "darkgreen"    ; fill land areas color
res@mpOceanFillColor        = "navy"         ; fill ocean areas color
res@mpInlandWaterFillColor  = "navy"         ; inland water color

plot = gsn_csm_vector_scalar_map(wks,u,v,mag,res) ; create the plot

end
```



NCL

Vector plot overlay on contour plot (1/3)

```
begin
  f   =  addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
  u   =  f->u10(0,:,:)                    ; 1st time step
  v   =  f->v10(0,:,:)                    ; 1st time step
  t   =  f->tsurf(0,:,:)                  ; 1st time step

  wks =  gsn_open_wks("png","plot_part_III_vector_colorize_overlay")

  cnres                    =  True        ; contour plot resources
  cnres@gsnDraw            =  False       ; don't draw plot, yet
  cnres@gsnFrame           =  False       ; don't advance frame

  cnres@gsnLeftString      =  t@long_name ; change left string
  cnres@gsnRightString     =  t@units     ; change right string

  cnres@tiMainString = \
                 "Overlay vectors on surface temperature contour plot"
```


NCL

Vector plot overlay on contour plot (2/3)

```
  cnres@cnFillOn       =  True            ; draw filled contours
  cnres@cnLinesOn      =  False           ; don't draw contour lines

  cnres@mpFillOn       =  False           ; no map fill

  vcres                =  True            ; vector resources
  vcres@gsnDraw        =  False           ; don't draw vector plot, yet
  vcres@gsnFrame       =  False           ; don't advance frame
  vcres@gsnRightString = " "              ; no right string
  vcres@gsnLeftString  = " "              ; no left string

  vcres@tiXAxisString  = " "              ; don't draw x-axis labels

  vcres@vcGlyphStyle     = "CurlyVector"  ; use curly vectors
  vcres@vcRefMagnitudeF =  20             ; define vector ref magnitude
  vcres@vcRefLengthF     =  0.045         ; define length of ref vector
  vcres@vcRefAnnoOrthogonalPosF = -1.36   ; move ref vector into plot
  vcres@vcRefAnnoString2On  =  False      ; don't label below ref vector
```

NCL

Vector plot overlay on contour plot (3/3)

```
cont_plot  =  gsn_csm_contour_map(wks,t,cnres)    ; contour plot
vec_plot   =  gsn_csm_vector(wks,u,v,vcres)       ; vector plot

overlay(cont_plot, vec_plot)          ; overlay vector on contour plot

draw(cont_plot)                                   ; draw the plot now
frame(wks)                                        ; advance the frame

end
```



NCL

Task: colorized vectors on contour plot

- Use the script `part_III_Vectors_colorize_overlay.ncl`
- Modify:
  - Use **variable** slp instead of tsurf; devide it by 100 to get **"hPa"** and change the **units attribute**
  - Use **colormap** "WhiteBlue" for contouring the **slp** variable
  - Select sub-region **lon**: -75-25º and **lat**: 30-85º
  - Don't draw the outer boxes of the labelbar
  - Set the **vector color** to "darkolivegreen3"
  - Set reference vector **magnitude** to 20 and **length** to 0.045
  - Make **vectors thicker** (=2.0)
  - Move the reference vector annotation into the **lower right corner** of the plot
  - Plot the reference arrow in **black**
  - Don't draw the string below the reference vector annotation

Hints:
- mpLimitMode, mpMinLonF,....
- cnLabelBarEndStyle
- vcRefAnnoOrthogonalPosF
- vcRefAnnoString2On
- vcRefAnnoArrowUseVecColor
- vcRefAnnoFontHeightF

NCL

Task: colorized vectors on contour plot



NCL

Task: colorized vectors on contour plot (1/4)

```
begin

   f = addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
   u   =  f->u10(0,:,:)                    ; first time step
   v   =  f->v10(0,:,:)                    ; first time step
   t   =  f->tsurf(0,:,:)                  ; first time step
   p   =  f->slp(0,:,:)                    ; first time step
   p   =  p/100                            ; convert to hPa units
   p@units  =  "hPa"                       ; set new units attribute

   wks =  gsn_open_wks("png","task_III_vectors")

;-- common resources for contour and vector plot
   res               =  True
   res@gsnDraw       =  False              ; don't draw plot, yet
   res@gsnFrame      =  False              ; don't advance frame
   res@tiMainString  = \
                "Task: Overlay vectors on contour plot of variable slp"
```


NCL

Task: colorized vectors on contour plot (2/4)

```
;-- set contour plot resources

   cnres                 =  res            ; contour resources
   cnres@cnFillOn        =  True           ; turn on fill colors
   cnres@cnFillPalette   = "WhiteBlue"     ; choose colormap
   cnres@cnLineThicknessF =  0.5      ; draw the contour lines thinner
   cnres@cnLabelBarEndStyle = "ExcludeOuterBoxes" ; don't draw outer
                                           ; labelbar boxes

   cnres@mpFillOn        =  False          ; turn off map fill
   cnres@mpLimitMode     = "LatLon"        ; map limit mode
   cnres@mpMinLonF       = -75             ; minimum longitude
   cnres@mpMaxLonF       =  25             ; maximum longitude
   cnres@mpMinLatF       =  30             ; minimum latitude
   cnres@mpMaxLatF       =  85             ; maximum latitude
```


NCL

Task: colorized vectors on contour plot (3/4)

```
   vcres                 =  res
   vcres@gsnRightString  = " "             ; no right string
   vcres@gsnLeftString   = " "             ; no left string

   vcres@vcGlyphStyle    = "CurlyVector"   ; use curly vectors
   vcres@vcLineArrowColor  = "darkolivegreen3"  ; set color for vectors
   vcres@vcLineArrowThicknessF  =  2.0          ; make vectors thicker

   vcres@vcRefMagnitudeF   =  20           ; vector ref. magnitude
   vcres@vcRefLengthF      =  0.045        ; length of ref. vector
   vcres@vcRefAnnoOrthogonalPosF = -0.452  ; move ref. vector into plot
   vcres@vcRefAnnoString2On      =  False  ; don't draw ref. annotation
   vcres@vcRefAnnoArrowUseVecColor =  False  ; use foreground color
   vcres@vcRefAnnoFontHeightF    = 0.008   ; font size ref. annotation
```

Task: colorized vectors on contour plot (4/4)

```
;-- create the plots, but don't draw them

   cont_plot  = gsn_csm_contour_map(wks,p,cnres)
   vec_plot   = gsn_csm_vector(wks,u,v,vcres)

;-- overlay vec_plot on cont_plot, but don't draw them

   overlay(cont_plot, vec_plot)

;-- now, draw the plot and advance frame

   draw(cont_plot)
   frame(wks)

end
```