NCL



Part V

Panel Plots

Exercises and Tasks

NCL

Panel Plots



NCL

Simple panel plot (1/2)

```
begin

  f  = addfile("$NCL_TUT/data/rectilinear_grid_3D.nc","r")
  var = f->tsurf

  wks = gsn_open_wks("png" ,"plot_part_V_simple_panel")

  res                =  True
  res@gsnDraw        =  False     ; don't draw the plot - yet
  res@gsnFrame       =  False     ; don't advance the frame

  res@cnFillOn       =  True      ; enable color filled contours
  res@cnLineLabelsOn =  False     ; don't draw labels on contour lines

  plot = new(3, "graphic")        ; assign empty plot array
```

Simple panel plot (2/2)

```
  res@tiMainString = "This is the 1st plot"            ; title string
  plot(0) = gsn_csm_contour_map(wks,var(0,:,:),res)    ; create plot 0

  res@tiMainString = "This is the 2nd plot"            ; title string
  plot(1) = gsn_csm_contour_map(wks,var(9,:,:),res)    ; create plot 1

  res@tiMainString = "This is the 3rd plot"            ; title string
  plot(2) = gsn_csm_contour_map(wks,var(19,:,:),res)   ; create plot 2

;-- create and plot the panel plot
  gsn_panel(wks,plot,(/3,1/),False)           ; rows = 3 ; columns = 1

 end
```


Simple panel plot



Panel plot with common labelbar (1/3)

```
begin
  f   = addfile("$NCL_TUT/data/rectilinear_grid_2D.nc","r")
  var = f->tsurf

  wks = gsn_open_wks("png" ,"plot_part_V_panel_attachment_one_lbar")

  res                  =  True
  res@gsnDraw          =  False        ; don't draw the plot, yet
  res@gsnFrame         =  False        ; don't advance the frame

  res@cnFillOn         =  True         ; enable color filled contours
  res@cnFillPalette    = "BlWhRe"      ; choose colormap
  res@cnLineLabelsOn   =  False        ; don't draw line labels
  res@cnInfoLabelOn    =  False        ; don't draw info label

  res@cnLevelSelectionMode = "ManualLevels" ; set contour level mode
  res@cnMinLevelValF   =  220          ; contour level minimum
  res@cnMaxLevelValF   =  315          ; contour level maximum
  res@cnLevelSpacingF  =    5          ; contour level interval
```

NCL

Panel plot with common labelbar (2/3)

```
  res@lbLabelBarOn     =  False        ; don't draw a labelbar

;-- create empty graphic plot array
  plot = new(3,"graphic")

;-- create the plots
  do i=0,2
      res@gsnCenterString   =  "t=" + (i+8)    ; draw center string
      plot(i) = gsn_csm_contour_map(wks,var((i+8),:,:),res)
  end do

;-- panel resources
  pres                  =  True
  pres@gsnPanelTop      =  0.96      ; panel top position (y-value)
  pres@gsnPanelBottom   =  0.012     ; panel bottom position (y-value)
  pres@gsnPanelLabelBar =  True      ; draw a common labelbar

  pres@lbLabelFontHeightF =  0.007   ; smaller labelbar font size
```


NCL

Panel plot with common labelbar (3/3)

```
  pres@txString        = \
     "Panel:  3 rows x 1 column + common legend and title"
                                   ; draw a common title on top
  pres@txFontHeightF   =  0.020    ; text font size
  pres@txPosXF         =  0.5      ; text x-position
  pres@txPosYF         =  0.97     ; text y-position
  pres@txJust          = "CenterCenter"  ; text justification

  gsn_panel(wks,plot,(/3,1/),pres)       ; draw the panel

end
```


NCL

Panel plot with common labelbar



NCL

Task: panel plot 3 rows x 3 columns

- Use the script `part_V_Panel_simple.ncl`
- Modify:
  - Use **colormap** "rainbow"
  - **Contour levels** from 220° to 315° K in 5° steps
  - Create plots of **time indexes** 2, 6, 10, 14, 18, 22, 26, 30, 34
  - Create panel plot with 3 rows **and** 3 columns
  - Draw a **common labelbar** and move it downward

Hints:
See part_V_panel_attachment_one_labelbar.ncl
- cnLevelSelectionMode
- cnMinLevelValF / cnMaxLevelValF
- cnLevelSpacingF
- lbLabelBarOn
- gsnPanelLabelBar
- gsnPanelTop / gsnPanelBottom



NCL

Task: panel plot 3 rows x 3 columns



NCL

Task: panel plot 3 rows x 3 columns (1/3)

```
begin
  f   = addfile("../../data/rectilinear_grid_2D.nc","r")
  var = f->tsurf

  wks = gsn_open_wks("png" ,"tasks_V_panel_3x3_one_labelbar")

  res                    =  True
  res@gsnDraw            =  False          ; don't draw the plot, yet
  res@gsnFrame           =  False          ; don't advance the frame
  res@gsnStringFontHeightF =  0.015        ; set string font size

  res@cnFillOn           =  True           ; enable color filled contours
  res@cnFillPalette      = "rainbow"       ; choose a colormap

  res@cnLevelSelectionMode = "ManualLevels"   ; set contour level mode
  res@cnMinLevelValF     =  220            ; contour level minimum
  res@cnMaxLevelValF     =  315            ; contour level maximum
  res@cnLevelSpacingF    =    5            ; contour level interval
```

Task: panel plot 3 rows x 3 columns (2/3)

```
  res@cnLineLabelsOn    =  False     ; don't draw labels on contour lines
  res@cnInfoLabelOn     =  False     ; don't draw labels on contour lines

  res@lbLabelBarOn      =  False     ; don't draw a labelbar

;-- assign empty graphic plot array
  nplots = 9
  plot = new(nplots,"graphic")

;-- create the plots
  do i=0,nplots-1
     if (i.ne.0) then
        res@gsnLeftString  = ""      ; only draw left string for plot(0)
        res@gsnRightString = ""      ; only draw right string for plot(0)
     end if
     m=2+(i*4)                    ; select every 4th time step, start t=2
     res@gsnCenterString   = "t=" + m
     plot(i) = gsn_csm_contour_map(wks,var(m,:,:),res)    ; create plots
  end do
```


Task: panel plot 3 rows x 3 columns (3/3)

```
;-- panel resources
  pres                =  True
  pres@gsnPanelTop    =  0.96         ; panel top position (y-value)
  pres@gsnPanelBottom  =  0.012       ; panel bottom position (y-value)
  pres@gsnPanelLabelBar =  True       ; draw a common labelbar

  pres@pmLabelBarOrthogonalPosF = -0.02     ; move labelbar downward

  pres@txString= "Panel:  3 rows x 3 column + common labelbar and title"
                                      ; draw a common title on top
  pres@txFontHeightF   =  0.020       ; text font size
  pres@txPosXF         =  0.5         ; text x-position
  pres@txPosYF         =  0.85        ; text y-position
  pres@txJust          = "CenterCenter"     ; text justification

  gsn_panel(wks,plot,(/3,3/),pres)

end
```